GRA 6227 – Modeling problems – Solutions

Problem A

1. Decision variables:

   X_1 : Number of units to be produced of product 1.

   X_2 : Number of units to be produced of product 2.

   Objective function:

   Maximize Totalprofit = $6 * X_1 + 4 * X_2$

   Constraints:

   $10 * X_1 + 10 * X_2 \leq 100$

   $7 * X_1 + 3 * X_2 \leq 42$

   $X_1 \geq 0$

   $X_2 \geq 0$

   The mathematical model on compact (set-based) form:

   Maximize totalprofit = $\sum_j p_t \cdot X_j$

   Subject to

   $\sum_j a_{i,j} \cdot X_j \leq c_i$ for each *i*

   $X_j \geq 0$ for each *j*

2. The optimal solution can be found for example by using AMPL.

Detailed model:

```
var X1 >= 0;
var X2 >= 0;

maximize totalprofit: 6*X1 + 4*X2;

s.t. Constraint1: 10*X1 + 10*X2 <= 100;
s.t. Constraint2:  7*X1 +  3*X2 <= 42;
```

```
ampl: solve;
MINOS 5.51: optimal solution found.
2 iterations, objective 46
ampl: display X1, X2;
X1 = 3
X2 = 7
```

Set-based model:

```
set PRODUCTS;
set LINES;
var X{PRODUCTS} >= 0;
param p{PRODUCTS};
param a{LINES, PRODUCTS};
param b{LINES};

maximize totalprofit:
  sum{j in PRODUCTS} p[j] * X[j];
subject to CapacityConstraints {i in LINES}:
  sum{j in PRODUCTS} a[i,j]*X[j] <= b[i];

data;
set PRODUCTS := P1 P2;
set LINES := Line1 Line2;
param p := P1 6  P2 4;
param a :    P1    P2  :=
      Line1  10    10
      Line2   7     3;
param b :=  Line1 100  Line2 42;
```

```
MINOS 5.51: optimal solution found.
2 iterations, objective 46
ampl: display X;
X [*] :=
P1  3
P2  7
```

Problem B

AMPL model:

```
set PRODUCTS;
set LINES;
var X{PRODUCTS} >= 0;
param price{PRODUCTS};
param varcost{PRODUCTS};
param fixedcost;
param minimumQuantity{PRODUCTS};
param maximumQuantity{PRODUCTS};

minimize totalvarcosts:
  sum{j in PRODUCTS} varcost[j] * X[j];

subject to MinimumRestriction {j in PRODUCTS}:
  X[j] >= minimumQuantity[j];

subject to MaximumRestriction {j in PRODUCTS}:
  X[j] <= maximumQuantity[j];

subject to BreakEven:
  sum{j in PRODUCTS} price[j]*X[j]
- sum{j in PRODUCTS} varcost[j]*X[j]
  = fixedcost;

data;
set PRODUCTS :=     Bass Ski Speed;
param price :=      Bass 23000   Ski 18000    Speed 26000;
param varcost :=    Bass 12500   Ski  8500    Speed 13700;
param fixedcost := 2800000;
param minimumQuantity := Bass  70  Ski  50  Speed  50;
param maximumQuantity := Bass 120  Ski 120  Speed 120;
```

Solution to AMPL model:

```
ampl: solve;
MINOS 5.51: optimal solution found.
4 iterations, objective 2925284.553
ampl: display X;
X [*] :=
 Bass   70
  Ski  120
Speed   75.2033
```

To break even, produce 70 of Bass, 120 of Ski, 75.2 of Speed.
(In practice, 75 units of Speed.)

Total sales revenues	5725285
Total variable costs	2925285
Total profit	5725285 – 2925285 – 2800000 = 0

Problem C

AMPL model:

```
set PRODUCTS;
var X{PRODUCTS} >= 0;
param price{PRODUCTS};
param cost{PRODUCTS};
param budget;
param demand{PRODUCTS};
param capacity;

maximize totalprofit:
  sum{j in PRODUCTS} (price[j] - cost[j]) * X[j];

subject to BudgetConstraint:
  sum{j in PRODUCTS} cost[j] * X[j] <= budget;

subject to CapacityConstraint:
  sum{j in PRODUCTS} X[j] <= capacity;

subject to MaximumSales {j in PRODUCTS}:
  X[j] <= demand[j];

data;
set PRODUCTS :=   Yodel Shotz RW;
param price :=    Yodel 3.00  Shotz 2.50  RW 1.75;
param cost :=     Yodel 1.50  Shotz 0.90  RW 0.50;
param budget :=   2000;
param demand :=   Yodel 400  Shotz 500  RW 300;
param capacity := 1000;
```

Solution to AMPL model:

```
ampl: solve;
MINOS 5.51: optimal solution found.
3 iterations, objective 1525
ampl: display X;
X [*] :=
   RW  100
Shotz  500
Yodel  400
```

Total profit = $1525.