Location models

GRA 6227 Business Optimisation

26.09.2019 & 02.10.2019

Atle Nordli

Location models

ELSEVIER

European Journal of Operational Research 111 (1998) 423–447

EUROPEAN JOURNAL OF OPERATIONAL RESEARCH

Invited Review

Strategic facility location: A review

Susan Hesse Owen *, Mark S. Daskin

Department of Industrial Engineering and Management Sciences, Northwestern University, Evanston, IL 60208-3119, USA

Accepted 1 April 1998

- Why do we focus on location models?

  1. The modeling methods are good examples of basic modeling techniques in mathematical modeling

  2. Location analysis is one of the big application areas for optimisation / operations research

Applications of location models

- Location of factories and/or warehouses
  («Supply network design»)

- Location of public services:
  Fire stations, Police offices, Hospitals, Ambulances, Helicopters, ……

Supply Chain Analytics

A.T. Kearney is an industry leader in embedding advanced analytics into supply chain solutions.

- **Design.** We work with you to make more informed supply chain design decisions regarding network configuration, manufacturing and distribution footprint and capacity, and inventory placement along the supply chain. We aim to help you meet customer requirements better using optimization and simulation techniques and enabling supplier collaboration early in the design process.

NEWS in ENGLISH.no
Views and News from Norway

You are here: Home / News / Bunad brigade back battling for babies

Bunad brigade back battling for babies

May 14, 2019

Around 200 women dressed in their traditional Norwegian bunads were back in battle mode on Tuesday, assembling in front of the Parliament in Oslo to demand a halt to closure and consolidation of maternity wards at small hospitals around the country. They met some resistance, though, from a clinic chief who thinks bigger facilities can provide better service.

Protests against hospital relocation

Over 80 patients and employees of the Neuromuscular Pathology Centre in Valcele, Covasna County, protested yesterday against the plans to relocate the hospital in Intorsura Buzaului. According to Mediafax, the protesters temporarily blocked the national road going through the locality. "It was a spontaneous reaction from the patients and employees and the protests will continue in the following days too," Vasile Neagovici, President Sanitas

Analysis of police office locations in Norway

NOU Norges offentlige utredninger **2013:9**

Ett politi – rustet til å møte fremtidens utfordringer

Politianalysen

VG NYHETER

< Nyheter | Innenriks | Utenriks | Siste 48t

Her er kutt-listen for fremtidens #politinorge: 151 kontorer kan bli nedlagt

****Ordførere vil ikke finne seg i nedleggelse av lensmannskontorer**
****Tysnes-ordføreren: - Opprop kan bli opprør**

Models in Owen and Daskin (1998)

(1) • The *P*-median problem (1) – (6)
  - Locate *P* facilities to **minimise <u>average</u> distance** to demand nodes

(3) • The set covering problem (7) – (9)
  - **Minimise the number** of facilities that cover all demand nodes

(4) • The maximal covering problem (10) – (14)
  - Locate *P* facilities to **maximise the number** of demand nodes that are covered

(2) • The vertex *P*-center problem (15) – (21)
  - Locate *P* facilities to **minimise <u>maximum</u> distance** to demand nodes

- A (dynamic) multiobjective maximal cover problem (22) – (27)
  - Multiple periods

- The expected regret problem (28) – (34)
  - Scenario-based *P*-median model to handle uncertainties («stochastic programming»)

Example:
A network of nodes



1, 2, .., 7 are demand nodes

A, B, C are candidate nodes for locating facilities

Distances:

	A	B	C	Demand
1	35	40	59	400
2	11	35	35	700
3	15	36	33	900
4	46	38	61	800
5	18	5	23	1400
6	42	51	31	500
7	62	46	35	300

(1) The *P*-median problem:

Choose ***P*** of the candidates locations so that total weighted distance between demand nodes and the nearest facility is minimised



	A	B	C	Demand
1	35	40	59	400
2	11	35	35	700
3	15	36	33	900
4	46	38	61	800
5	18	5	23	1400
6	42	51	31	500
7	62	46	35	300

Total demand = **5000**

(1) The *P*-median problem:

For ***P*** = 2,
the optimal solution is to use A and B.
Nodes 1, 2, 3 , 6 are assigned to A.
Nodes 4, 5, 7 are assigned to B.
Total weighted distance: 107400 km
Average distance per individual:
107400 / 5000 = 21,4 km



	A	B	C	Demand
1	35	40	59	400
2	11	35	35	700
3	15	36	33	900
4	46	38	61	800
5	18	5	23	1400
6	42	51	31	500
7	62	46	35	300

The *P*-median problem as formulated in Owen and Daskin

Decision variables:

$$X_j = \begin{cases} 1 & \text{if we locate at potential facility site } j, \\ 0 & \text{if not.} \end{cases}$$

$$Y_{ij} = \begin{cases} 1 & \text{if demands at node } i \text{ are served by a facility at node } j, \\ 0 & \text{if not.} \end{cases}$$

Using these definitions, the *P*-median problem can be written as the following integer linear program:

$$\text{Minimize} \quad \sum_i \sum_j h_i d_{ij} Y_{ij} \tag{1}$$

$$\text{subject to:} \quad \sum_j X_j = P, \tag{2}$$

$$\sum_j Y_{ij} = 1 \quad \forall i, \tag{3}$$

$$Y_{ij} - X_j \leqslant 0, \quad \forall i, j, \tag{4}$$

$$X_j \in \{0, 1\} \quad \forall j, \tag{5}$$

$$Y_{ij} \in \{0, 1\} \quad \forall i, j. \tag{6}$$

Inputs:
i = index of demand node
j = index of potential facility site
h_i = demand at node i
d_{ij} = distance between demand node i and potential facility site j
P = number of facilities to be located

The *P*-median problem formulated in AMPL

```
set NODES; # Set of demand nodes
set CANDIDATES; # Set of POTENTIAL facilities
param P; # predetermined number of facilities to locate
param h{NODES}; # demand at demand node i
param d{NODES,CANDIDATES}; # distance between i and j
var X{CANDIDATES} binary; # locate a potential facility at site j (yes/no)
var Y{NODES,CANDIDATES} binary; # demand node i is served by facility at site j (yes/no)

minimize DemandWeightedDistance:
    sum{i in NODES, j in CANDIDATES} h[i]*d[i,j]*Y[i,j];

s.t. Constr2:
    sum{j in CANDIDATES} X[j] = P;      # Choose P facility locations
s.t. Constr3 {i in NODES}:
    sum{j in CANDIDATES} Y[i,j] = 1;    # Assign one facility per demand node:
s.t. Constr4 {i in NODES, j in CANDIDATES}:
    Y[i,j] <= X[j];                     # No assignment (i to j) if no facility at j:
```

```
data;
set NODES := D01, D02, D03, D04, D05, D06, D07;
set CANDIDATES := F1, F2, F3;
param P := 2;
param h := D01      400
           D02      700
           D03      900
           D04      800
           D05     1400
           D06      500
           D07      300 ;
```

```
param d:
        F1      F2      F3   :=
D01     35      40      59
D02     11      35      35
D03     15      36      33
D04     46      38      61
D05     18      5       23
D06     42      51      31
D07     62      46      35   ;
```

Comparison of math formulation and AMPL model

$$\boxed{\sum_j}\boxed{Y_{ij}} = 1 \quad \boxed{\forall i,} \qquad (3)$$

```
# Assign one facility per demand node:
s.t. Constr3 {i in NODES}:  sum{j = 1..J} Y[i,j] = 1;
```

Comparison of math formulation and AMPL model

$$Y_{ij} - X_j \leqslant 0, \quad \forall i, j, \qquad (4)$$

```
# No assignment (i to j) if no facility at j:
s.t. Constr4 {i = 1..I, j = 1..J}: Y[i,j] <= X[j];
```

(2) The vertex *P*-center problem :

Choose ***P*** of the candidates locations so that **maximum** distance between any demand node and its nearest facility is **minimised**



	A	B	C	Demand
1	35	40	59	400
2	11	35	35	700
3	15	36	33	900
4	46	38	61	800
5	18	5	23	1400
6	42	51	31	500
7	62	46	35	300

Total demand = **5000**

(2) The vertex *P*-center problem :

For ***P*** = 2,
the optimal solution is to use B and C. Nodes 1, 4, 5 are assigned to B.
Nodes 2, 3, 6, 7 are assigned to C.
Maximum distance: 40km (between 1 and B)

(Note that 2 assigned to B is an alternative optimal solution)



	A	B	C	Demand
1	35	40	59	400
2	11	35	35	700
3	15	36	33	900
4	46	38	61	800
5	18	5	23	1400
6	42	51	31	500
7	62	46	35	300

The vertex *P*-center problem as formulated in Owen and Daskin

The following additional decision variable is needed in order to formulate the vertex *P*-center problem:
$D =$ maximum distance between a demand node and the nearest facility.

The resulting integer programming formulation of the vertex *P*-center problem follows.

$$\text{Minimize} \quad D \tag{15}$$

$$\text{subject to:} \quad \sum_j X_j = P, \tag{16}$$

$$\sum_j Y_{ij} = 1 \quad \forall i, \tag{17}$$

$$Y_{ij} - X_j \leqslant 0 \quad \forall i, j, \tag{18}$$

$$D \geqslant \sum_j d_{ij} Y_{ij} \quad \forall i, \tag{19}$$

$$X_j \in \{0, 1\} \quad \forall j, \tag{20}$$

$$Y_{ij} \in \{0, 1\} \quad \forall i, j. \tag{21}$$

The vertex *P*-center problem formulated in AMPL

```
set NODES; # Set of demand nodes
set CANDIDATES; # Set of POTENTIAL facilities
param P; # predetermined number of facilities to locate
param h{NODES}; # demand at demand node i
param d{NODES,CANDIDATES}; # distance between i and j
var X{CANDIDATES} binary; # locate a potential facility at site j (yes/no)
var Y{NODES,CANDIDATES} binary; # demand node i is served by facility at site j (yes/no)
var D >=0 ;

minimize MaximalDistance:  D;
s.t. Constr16:
    sum{j in CANDIDATES} X[j] = P;     # Choose P facility locations
s.t. Constr17 {i in NODES}:
    sum{j in CANDIDATES} Y[i,j] = 1;  # Assign one facility per demand node
s.t. Constr18 {i in NODES, j in CANDIDATES}:
    Y[i,j] <= X[j];                   # No assignment (i to j) if no facility at j
s.t. Constr19 {i in NODES}:
    D >= sum{j in CANDIDATES} d[i,j]*Y[i,j];
```

```
data;
set NODES := D01, D02, D03, D04, D05, D06, D07;
set CANDIDATES := F1, F2, F3;
param P := 2;
param d:
        F1      F2      F3   :=
D01     35      40      59
D02     11      35      35
D03     15      36      33
D04     46      38      61
D05     18      5       23
D06     42      51      31
D07     62      46      35   ;
```

Comparison of math formulation and AMPL model

$$D \geqslant \sum_j d_{ij} Y_{ij} \quad \forall i, \qquad (19)$$

```
s.t. Constr19 {i in NODES}:
    D >= sum{j in CANDIDATES} d[i,j]*Y[i,j];
```

(3) The set covering problem :

A demand node is said to be **covered** if it is within a predefined distance from the nearest facility.

For each facility, there is **fixed cost**.

Choose the portfolio of facility locations such that **all** demand nodes are **covered.**



Before the optimisation, a «cover matrix» must be generated, based on a given maximum distance (here: 50km)

Distance matrix

	A	B	C
1	35	40	59
2	11	35	35
3	15	36	33
4	46	38	61
5	18	5	23
6	42	51	31
7	62	46	35

Cover matrix (50km)

	A	B	C
1	1	1	0
2	1	1	1
3	1	1	1
4	1	1	0
5	1	1	1
6	1	0	1
7	0	1	1

(3) The set covering problem:

Assume fixed costs:
$c_A = 3000$,
$c_B = 4000$,
$c_C = 2000$.
Then, the optimal solution is to use A and C.

Total costs = 3000 + 2000 = 5000



Cover matrix (50km)

	A	B	C
1	1	1	0
2	1	1	1
3	1	1	1
4	1	1	0
5	1	1	1
6	1	0	1
7	0	1	1

How to create cover matrix in Excel

	A	B	C	D	E	F	G	H	I	J	K	L	M
1							max distance						
2			A	B	C		50			A	B	C	
3		1	35	40	59				1	1	1	0	
4		2	11	35	35				2	1	1	1	
5		3	15	36	33				3	1	1	1	
6		4	46	38	61				4	1	1	0	
7		5	18	5	23				5	1	1	1	
8		6	42	51	31				6	1	0	1	
9		7	62	46	35				7	0	1	1	
10													
11						Formula in J3:	=IF(C3<=$G$3;1;0)						
12						Formula in J4:	=IF(C4<=$G$3;1;0)						
13							Etc...						

The set covering problem as formulated in Owen and Daskin

Inputs:
$c_j =$ fixed cost of siting a facility at node j
$S =$ maximum acceptable service distance (or time)
$N_i =$ set of facility sites j within acceptable distance of node i (i.e., $N_i = \{j | d_{ij} \leqslant S\}$)

The set covering problem can thus be represented by the following integer program:

$$\text{Minimize} \quad \sum_j c_j X_j \tag{7}$$

$$\text{subject to:} \quad \sum_{j \in N_i} X_j \geqslant 1 \quad \forall i, \tag{8}$$

$$X_j \in \{0, 1\} \quad \forall j. \tag{9}$$

The objective function (7) minimizes the cost of facility location. In many cases, the costs c_j are assumed to be equal for all potential facility sites j, implying an objective equivalent to minimizing the number of facilities located. Constraint (8) requires that all demands i have at least one facility located within the acceptable service distance. The remaining constraints (9) require integrality for the decision variables.

Constraint (8) in the Owen and Daskin model can be formulated like this

Minimize	$\sum_j c_j X_j$		(7)
subject to:	$\boxed{\sum_j a_{i,j} X_j \geq 1}$	for all *i*	(8b)
	$X_j \in \{0,1\}$	for all *j*	(9)

$a_{i,j}$ **is the 0/1 cover matrix (a parameter),**
which we can generate in Excel prior to the optimization

The set covering problem formulated in AMPL

```
set NODES; # Set of demand nodes
set CANDIDATES; # Set of POTENTIAL facilities
param c{CANDIDATES}; # fixed cost for candidate location j
param a{NODES,CANDIDATES}; # the cover matrix
var X{CANDIDATES} binary; # locate a potential facility at site j (yes/no)

minimize TotalCosts:  sum{j in CANDIDATES} c[j]*X[j];
s.t. Constr8b {i in NODES}: sum{j in CANDIDATES} a[i,j]*X[j] >= 1;

data;
set NODES := D01, D02, D03, D04, D05, D06, D07;
set CANDIDATES := F1, F2, F3;
param c := F1 3000  F2 4000  F3 2000;
param a:
        F1      F2      F3   :=
D01     1       1       0
D02     1       1       1
D03     1       1       1
D04     1       1       0
D05     1       1       1
D06     1       0       1
D07     0       1       1   ;
```

The maximal covering problem as formulated in Owen and Daskin

Specifically, the maximal covering problem seeks to maximize the amount of demand covered within the acceptable service distance S by locating a fixed number of facilities. The formulation of this problem requires the following additional set of decision variables:

$$Z_i = \begin{cases} 1 & \text{if node } i \text{ is covered,} \\ 0 & \text{if not.} \end{cases}$$

Combining these variables with the notation defined above, we derive the following formulation of the maximal covering problem:

$$\text{Maximize} \quad \sum_i h_i Z_i \tag{10}$$

$$\text{subject to:} \quad Z_i \leqslant \sum_{j \in N_i} X_j \quad \forall i, \tag{11}$$

$$\sum_j X_j \leqslant P, \tag{12}$$

$$X_j \in \{0, 1\} \quad \forall j, \tag{13}$$

$$Z_i \in \{0, 1\} \quad \forall i. \tag{14}$$

Constraint (11) in the Owen and Daskin model can be formulated like this

Maximize	$\sum_i h_i Z_i$		(10)
subject to:	$\boxed{Z_i \leq \sum_j a_{i,j} X_j}$	for all *i*	(11b)
	$\sum_j X_j \leq P$		(12)
	$X_j \in \{0,1\}$	for all *j*	
	$Z_i \in \{0,1\}$	for all *i*	

$a_{i,j}$ **is the 0/1 cover matrix (a parameter),**
which we can generate in Excel prior to the optimization

Task

- At Itslearning, there is a Word file containing data for the Maximal Covering problem.
  - 20 candidate locations
  - 90 demand locations
  - Total demand = 1420401
- You can copy and paste these data into AMPL.
- Then, based on the given data and the formulation in the previous slide, build a model file for the Maximal Covering problem and solve it.
  - Which locations should be chosen if ***P*** = 6?
  - How much of the total demand will then be covered?
  - How much is more of the total demand wil be covered if ***P*** = 7?